ECON 331

ASSIGNMENT #3

QUESTION 1

Typically, one would not use Cramer's rule in Maple to solve for variables in a system of linear equations. Cramer's rule is a conveneint compuational device which allows us to solve linear equation sytems 'by hand'. In Maple we could find the inverse demand functions using the solve command as we did in assignment #1. Here is another way of solving linear systems using the linalg package:

```
> with(linalg):
```

The coefficient matrix is:

```
> A:=matrix([[2,-1],[-1,3]]);
```

$$A := \begin{bmatrix} 2 & -1 \\ -1 & 3 \end{bmatrix}$$

The vector of constants is:

```
> d:=vector([20-q1,25-q2]);
```

$$d := [20 - q1, 25 - q2]$$

The solution vector (p1, p2) is:

```
> multiply(inverse(A),d);
```

$$\left[17 - \frac{3}{5} q1 - \frac{1}{5} q2, 14 - \frac{1}{5} q1 - \frac{2}{5} q2 \right]$$

If you really wanted to use Cramer's rule, you could do something like this:

```
> A1:=augment(d,subvector(A,1..2,2));
```

$$A1 := \begin{bmatrix} 20 - q1 & -1 \\ 25 - q2 & 3 \end{bmatrix}$$

```
> A2:=augment(subvector(A,1..2,1),d);
```

$$A2 := \begin{bmatrix} 2 & 20 - q1 \\ -1 & 25 - q2 \end{bmatrix}$$

The inverse demand functions are:

```
> p1:=det(A1)/det(A);
> p2:=det(A2)/det(A);
```

$$p1 := 17 - \frac{3}{5} q1 - \frac{1}{5} q2$$

$$p2 := 14 - \frac{1}{5} q1 - \frac{2}{5} q2$$

QUESTION 2

a)

Here we will use a slightly different approach to implement Cramer's rule.

```
> restart;
> with(linalg):
Warning, the protected names norm and trace have been redefined and unprotected
```

From assignment #1, we know that the coefficient matrix for the IS-LM sytem is:

```
> A:=matrix(2, 2, [1-b+bt, a, k, -B]);
```

$$A := \begin{bmatrix} 1 - b + bt & a \\ k & -B \end{bmatrix}$$

The vector of constants is:

```
> d:=matrix(2,1,[Co+Io+G, M0]);
```

$$d := \begin{bmatrix} Co + Io + G \\ M0 \end{bmatrix}$$

We can find Ye :

```
> AY:=copyinto(d,copy(A),1,1);
```

$$AY := \begin{bmatrix} Co + Io + G & a \\ M0 & -B \end{bmatrix}$$

```
> Ye:=det(AY)/det(A);
```

$$Ye := \frac{-B\,Co - B\,Io - B\,G - a\,M0}{-B + B\,b - B\,bt - a\,k}$$

```
> simplify(Ye);
```

$$\frac{B\,Co + B\,Io + B\,G + a\,M0}{B - B\,b + B\,bt + a\,k}$$

```
> collect(%,B);
>
```

$$\frac{(Co + Io + G)\,B + a\,M0}{(1 - b + bt)\,B + a\,k}$$

Similarly, we can find re:

```
> Ar:=copyinto(d,copy(A),1,2);
```

$$Ar := \begin{bmatrix} 1 - b + bt & Co + Io + G \\ k & M0 \end{bmatrix}$$

```
> re:=det(Ar)/det(A);
```

$$re := \frac{M0 - M0\,b + M0\,bt - k\,Co - k\,Io - k\,G}{-B + B\,b - B\,bt - a\,k}$$

b)

The coefficient on M0 is:

```
> coeff(Ye,M0,1);
```

$$-\frac{a}{-B + B\,b - B\,bt - a\,k}$$

```
> simplify(%);
```

$$\frac{a}{B - B\,b + B\,bt + a\,k}$$

```
> collect(%,B);
```

$$\frac{a}{(1 - b + bt)\,B + a\,k}$$

```
>
```

Given the standard assumptions about the magnitude of the parameters in the expression above, the coefficient on M0 is postive. This means that an increase in the exogenous stock of money will 'cause' an unambiguous increase in Ye.